

GEOGRAPHIC INFORMATION SYSTEM PROGRAMS FOR USE IN THE WATER-SUPPLY- ALLOCATION PERMITTING PROCESS

by Paul Dunne and Curtis V. Price

U.S. GEOLOGICAL SURVEY
Open-File Report 95-157



Prepared in cooperation with the
NEW JERSEY DEPARTMENT OF ENVIRONMENTAL PROTECTION

West Trenton, New Jersey

1995

U.S. DEPARTMENT OF THE INTERIOR

BRUCE BABBITT, Secretary

U.S. GEOLOGICAL SURVEY

Gordon P. Eaton, *Director*

For additional information write to:

U.S. Geological Survey
Mountain View Office Park
Suite 206
810 Bear Tavern Road
West Trenton, NJ 08628

Copies of this report can be purchased from:

U.S. Geological Survey
Earth Science Information Center
Open-File Reports Section
Denver Federal Center
Box 25286, MS 517
Denver, CO 80225

CONTENTS

	Page
Abstract.....	1
Introduction.....	1
Purpose and scope.....	2
Overview of the geographic information system.....	2
Command language.....	2
User interface.....	3
Geographic information system programs for use in the water- supply-allocation permitting process.....	3
Main menu.....	4
Start.aml program.....	4
Design.menu program.....	4
Location and draw features.....	4
Llfind.aml program.....	7
Select.menu program.....	7
Screen and hardcopy plots.....	8
Order.aml program.....	8
Plot.aml program.....	8
Zoomout.aml program.....	8
Plotfile.aml program.....	9
Plot2.aml program.....	9
Surface-water and ground-water sites.....	9
Calcgw.aml and Calcsw.aml programs.....	9
Glist.aml and Slist.aml programs.....	9
Terminal options.....	11
References cited.....	13
Appendix 1. Source code for start.aml.....	14
2. Source code for design.menu.....	16
3. Source code for llfind.aml.....	18
4. Source code for select.menu.....	20
5. Source code for order.aml.....	22
6. Source code for plot.aml.....	23
7. Source code for zoomout.aml.....	24
8. Source code for plotfile.aml.....	25
9. Source code for plot2.aml.....	28
10. Source code for calcgw.aml and calcsw.aml.....	29
11. Source code for glist.aml and slist.aml.....	30

ILLUSTRATIONS

Figure 1. Example menu and the associated command file.....	3
2. Main menu of the geographic information system programs and sections.....	6
3. Example of the coverage menu of the geographic information system program.....	7
4. Example of a hardcopy plot generated by the geographic information system.....	10

TABLES

Table 1. Data bases and selected attributes.....	5
2. Attribute list associated with selected ground-water- withdrawal sites, generated by the geographic information system.....	12

CONVERSION FACTORS

<u>Multiply</u>	<u>BY</u>	<u>To obtain</u>
gallon (gal)	3.785	liter
mile (mi)	1.609	kilometer

GEOGRAPHIC INFORMATION SYSTEM PROGRAMS FOR USE IN THE WATER-SUPPLY-ALLOCATION PERMITTING PROCESS

by Paul Dunne and Curtis V. Price

ABSTRACT

This report describes computer programs designed for use with a geographic information system as an aid in determining the suitability of proposed ground-water and surface-water withdrawal sites. These programs enable a user to display proposed water-supply-allocation sites in a defined area together with present water-supply-allocation sites and important hydrologic and geographic features on a computer screen or on hardcopy plots. The programs are menu-driven and do not require familiarity with geographic information systems. Source codes for the programs are included in appendixes.

INTRODUCTION

Water-supply demands continue to increase as a result of residential and industrial development. New water-supply-allocation sites are needed to meet this demand. Proposed water-supply-allocation sites are evaluated to determine their effect on the present sites and the local water supply. A variety of geographic and hydrologic features are examined to determine whether a proposed site should be approved for future allocations.

During the approval process, maps showing proposed and present water-supply-allocation sites and geographic and hydrologic features are produced. Proposed water-supply-allocation sites are compared to present sites to determine whether they tap a common source or whether the proposed sites are vulnerable to contaminants, such as effluents from hazardous waste sites (Michael Bleicher, New Jersey Department of Environmental Protection, written commun., 1991).

The U.S. Geological Survey (USGS) in cooperation with the New Jersey Department of Environmental Protection (NJDEP) conducted a project to develop geographic information system (GIS) programs that produce plots showing proposed water-supply-allocation sites, present water-supply-allocation sites, and important hydrologic and geographic features within a defined area. To produce the plots, the GIS programs are used with a data base of surface-water- and ground-water-withdrawal sites that have water allocation permits. The plots can be used to evaluate the potential effects of proposed ground-water and surface-water withdrawals on local water resources.

The GIS programs produce a menu-based user interface which enables a user, who does not necessarily have an understanding of the programs or of GIS, to identify important information. The GIS menus enable a user to invoke computer commands by selecting menu options with the keyboard or computer mouse device. The GIS programs can access a greater amount of information more rapidly than current methods.

Purpose and Scope

This report describes GIS programs that can be used in conjunction with geographic and hydrologic data bases to facilitate the evaluation of the potential effects of additional surface-water and ground-water withdrawals on water supplies. The computer program source codes are documented in appendixes.

Overview of the Geographic Information System

A GIS analyzes data sets and displays the data as a map or composite image. The GIS programs discussed below were written with ARC/INFO¹ version 5.0 for use on a PRIME computer, although they can be used with ARC/INFO on other computer systems. ARC/INFO is a GIS software package that automates, manipulates, analyzes, and displays geographic data in digital form (Environmental Systems Research Institute, Inc., v.1, 1987a). ARCPLOT is the cartographic display program of ARC/INFO that creates maps, and INFO is the relational data-base manager for the tabular data associated with geographic features (Environmental Systems Research Institute, Inc., v.1, 1987a). INFO stores data on geographic and topographic features and their corresponding tabular records.

GIS can be used with a variety of digital data sets. The data sets used in this project are not an all-inclusive group. Any number of data sets can be used with GIS in one session, according to the user's needs and applications.

Command Language

ARC/INFO Macro Language (AML) was used to write the customized GIS programs. AML allows the user to design new ARC/INFO commands, run a series of ARC/INFO commands, run user menus, and develop user interfaces (Environmental Systems Research Institute Inc., 1987a). With AML, the user can write complex programs that accept input from the terminal or from system files and perform tasks by using both ARC/INFO and computer operating system commands.

¹ Use of trade, brand, or firm names in this report is for identification purposes only and does not constitute endorsement by the U.S. Geological Survey.

User Interface

A user interface, or AML menu, is a list of options consisting of keywords or phrases that are displayed on the terminal screen (Environmental Systems Research Institute Inc., 1987b). The user can select from the menu by moving the screen cursor to the position of the keyword or phrase and pressing the return key on the terminal keypad or by pointing with a mouse device and clicking a mouse button. This action will execute a command in ARC/INFO, such as display information, plot a map, or invoke another menu.

The menu enables the user to operate GIS software without learning complex commands. Each keyword or phrase is included in a text file that explicitly defines the menu choices and associates them with specific actions. An example menu and the associated command file are shown in figure 1.

Example menu	Associated command file
	2 menufile

Make plot	'Make plot' &menu plot.menu
Clear screen	'Clear screen' CLEAR
Quit	'Quit' QUIT

Figure 1. Example menu and the associated command file.

GEOGRAPHIC INFORMATION SYSTEM PROGRAMS FOR USE IN THE WATER-SUPPLY-ALLOCATION PERMITTING PROCESS

The customized GIS programs are designed as step-by-step methods for making decisions during the water-supply-allocation permitting process. GIS menus present choices and allow the user to display information by using the computer mouse device or the keypad. After selecting the information that is pertinent to the permitting decision, the user can produce hardcopy and screen plots of the selected information. The programs also can be used to create tables that list selected attributes associated with selected surface-water- and ground-water-withdrawal sites.

The GIS programs were developed to be used with available digital data bases. The primary data base used with the programs contains data on ground-water- and surface-water-withdrawal sites in the State. These data were obtained from the NJDEP (Michael Bleicher, New Jersey Department of Environmental Protection, written commun., 1991). The data base also includes unique site identification numbers, well depths, elevations of the land surface at ground-water-withdrawal sites, and location identifiers, such as latitude and longitude. Other data used in the GIS programs include stream hydrography, drainage-basin outlines, and political boundaries. Each type of data is stored in its own data base. A list of the data bases, their selected attributes, and the sources of the data are shown in table 1.

The programs discussed in this report can be copied and used on a computer system that has the capacity to run ARC/INFO. The programs cannot be used with the personal computer version of ARC/INFO.

Main Menu

The main menu of the geographic information system as it appears on the terminal screen is shown in figure 2. Main menu suboptions are additional menus that can be displayed by choosing from the main menu options. In this report, the main menu is divided into four sections--Location and draw features, Screen and hardcopy plots, Surface-water- and ground-water-withdrawal sites, and Terminal options--for discussion purposes.

Start.aml Program

This program must be run first to start the menu; the source code for start.aml is shown in appendix 1. In this program, the local computer pathname must be entered. To begin, the terminal and display devices should be set by the user as follows (All commands to be typed by the user are underlined.):

>: ARC

ARC: &STATION <STATION NAME> (The user replaces <STATION NAME> with the name of the workstation.)

ARC: &R START

ARC PLOT is activated, and the main menu of the GIS is displayed by invoking the program, design.menu.

Design.menu Program

The design.menu program (app. 2) displays the main menu of the GIS. This program is similar in structure to the associated text file shown in figure 1.

Location and Draw Features

The primary data associated with the location feature are the latitude and longitude of the proposed water-supply-allocation site. Radius information, which also is associated with the

Table 1. Data bases and selected attributes

[USGS, U.S. Geological Survey; NJDEP, New Jersey Department of Environmental Protection]

Data base	Selected attributes	Data type	Source
Ground-water-withdrawal sites and Surface-water-withdrawal sites	Watershed basin County Latitude Longitude Water allocation in millions of gallons per minute (Mgal/m) Permit number USGS identification number	Character Character Integer Integer Numeric Character Character	NJDEP computer files
Township	Township name County name	Character Character	USGS 1:24,000 topographic maps, digitized by NJDEP
County	County name	Character	USGS 1:24,000 topographic maps, digitized by NJDEP
Stream	None	None	USGS 1:250,000-scale quadrangle, digitized by USGS
Drainage basin	First order basin name Second order basin name Third order basin name	Character Character Character	USGS computer files
Pinelands area	None	None	NJDEP computer files

<u>Main menu options</u>	<u>Main menu suboptions</u>	<u>Sections</u> ¹
Set site LAT,LON,RAD Choose draw features		Location and draw features
Draw (mapextent)	First order basin Second order basin Third order basin Radius area	Screen and hardcopy plots
Zoom in/out	Zoom in box Zoom out	
Set Mapscale	automatic 1:24,000	
Make a plotfile		
Identify	GW site SW site Basin	
Refresh screen		
Select GW area	upstream polygon buffer area mapextent clr all selections	Surface-water- and ground-water- withdrawal sites
Create GW data list	list1 ² list2 ²	
List GW data to screen --	list1 ² list2 ²	
Select SW area	upstream polygon buffer area mapextent clr all selections	
Create SW data list	list1 ² list2 ²	
List SW data to screen --	list1 ² list2 ²	
Clear Screen Quit menu		Terminal options

¹Sections do not appear on the computer screen.

²List1 and list2 can be replaced with a name supplied by the user, according to information to be produced for a hardcopy plot. This subsection also can be expanded to include additional list options.

Figure 2. Main menu of the geographic information system programs and sections.

location feature, is designed to identify existing surface-water-or ground-water-withdrawal sites within an assigned radius of the proposed site. Water-supply-allocation permit data also will be accessed from the location feature. Draw features are associated with the digital data sets, such as streams, drainage basins, townships, and counties (table 1). Location and draw features must be chosen by the user, interactively with the computer mouse device or keypad, before proceeding to other menu selections.

Llfind.aml Program

The llfind.aml program (app. 3) allows the user to select information that is associated with the specified radius of a proposed site from the data bases; it is run when the Set site LAT, LON, RAD selection is chosen from the main menu. This program allows the user to access a specific location, such as a proposed water-supply-allocation site, by typing in a latitude and longitude. The radius for this locational point also must be entered (for example, 5.3 miles).

Select.menu Program

The main menu option Choose draw features invokes the select.menu program (app. 4). After selecting Choose draw features, the user is prompted with a menu that identifies the coverages to be drawn. A coverage is an ARC/INFO map data set of spatial and tabular information (Environmental Systems Research Institute, Inc., v.1, 1987). This menu as it appears on the terminal screen is shown in figure 3. The user must answer yes or no to all coverage choices given in this menu. A yes or y response will result in a screen image or a hardcopy plot of the coverage. Select.menu can be expanded to accommodate additional data sets for screen display or hardcopy plot.

Select coverages to be drawn:
(Answer "yes", "no", "y" or "n" for ALL coverages.)

<u>Coverage</u>	<u>On?</u>
Townships	
Counties	
Pinelands	
Streams (250k)	
Basins	
Surface-water withdrawal points	
Ground-water withdrawal points	

OK

Figure 3. Example of the coverage menu of the geographic information system program.

Screen and Hardcopy Plots

Options in the Screen and hardcopy plots section of the main menu produce plots on the computer screen or on hardcopy (fig. 2). Four ARC/INFO commands are used in this section--mapextent, mapscale, identify, and clear. Mapextent delineates the area to be shown on the screen or on hardcopy; mapscale specifies the scale at which the plot will be drawn; identify lists specific attributes of a feature selected on the screen or on hardcopy, such as ground-water site, surface-water site, and drainage basin; and clear refreshes the terminal screen, allowing the user to redraw a plot. (See app.2 for specific locations of these commands.) Five AML programs have been prepared that will produce screen and hardcopy plots--order.aml, plot.aml, zoomout.aml, plotfile.aml, and plot2.aml.

Order.aml Program

The Draw (mapextent) option of the main menu allows the user to create a screen plot or hardcopy by defining an area on the computer screen. The Draw (mapextent) option contains four suboptions--First order basin, Second order basin, Third order basin, and Radius area. The first three suboptions, which run the program order.aml (app. 5), allow the user to choose a first, second, or third order basin from a list of drainage basins. (Each drainage basin is classified according to the surface-water body that drains it.) All draw features that were selected in the Location and draw feature section will be plotted. The fourth suboption, Radius area, uses the mapextent command to plot a map with the draw features that were chosen in the Location and draw features section, including the specified radius.

Plot.aml Program

The plot.aml program (app. 6), which is designed to draw an outline of the basins within the mapextent area to the terminal screen, is activated when Draw (mapextent) is chosen from the main menu. Draw (mapextent) suboptions further define the information to be produced on the terminal screen.

Zoomout.aml Program

The Zoom in/out option of the main menu presents two suboptions--Zoom in box and Zoom out. Zoom in box allows the user to enlarge any percentage of the area shown on the terminal screen by using the computer mouse device or keypad. The Zoom out option runs the zoomout.aml program (app. 7), which increases the mapextent by 20 percent and draws the new plot on the screen.

Plotfile.aml Program

The Make a plotfile option produces a hardcopy of the most recent plot produced on the terminal screen by the plot.aml program. The user will be prompted for a plotfile name; any name can be used. A typical hardcopy plot is shown in figure 4. The source code for plotfile.aml is listed in appendix 8.

Plot2.aml Program

The plot2.aml program (app. 9) is a module within plotfile.aml. Plot2.aml contains information, such as colors and symbols, to be used by the plotter when producing a hardcopy.

Surface-Water and Ground-Water Sites

The third section of the main menu enables the user to identify existing surface- and ground-water-withdrawal sites by accessing and presenting information or attributes from within the specified radius that previously has been defined and drawn.

Calcgw.aml and Calcs.aml Programs

Distances between water-supply-allocation points can be determined by choosing the Select GW area or Select SW area option. (GW means ground water, and SW means surface water.) Each option has three suboptions which can run either calcgw.aml for ground water or calcs.aml for surface water (app. 10). The three suboptions--upstream polygon, buffer area, and mapextent--are programs that calculate the distance, in miles, between a proposed site (chosen with llfind.aml) and existing surface-water- or ground-water-withdrawal sites within the specified area.

The upstream polygon suboption enables the user to define an area of any size with the computer mouse device in order to determine the distance between sites. The user can choose to examine only those ground-water or surface-water sites within a specified radius (defined in llfind.aml) by selecting the buffer area suboption. If the entire area within the existing mapextent is desired, the mapextent suboption is selected. The clr all selections suboption clears the selected suboption so that other selections can be made.

Glist.aml and Slist.aml Programs

After selecting a set of withdrawal sites, the user can produce a hardcopy list of various attributes associated with these sites by choosing Create GW data list or Create SW data list from the main menu. The suboptions are upstream polygon, buffer area, mapextent, and clr all selections (fig. 2). Glist.aml and slist.aml programs for ground-water sites and surface-water sites, respectively, (app. 11) produce lists of attributes and save them in ASCII computer files. If the file is




Figure 4. Example of hardcopy plot generated by the geographic information system.

to be shown on the screen, choose List GW data to screen or List SW data to screen from the main menu options. A hardcopy can be produced by printing the ASCII file from the computer after leaving ARC/INFO (table 2).

Terminal Options

The terminal options section of the main menu does not contain any programs. The options that run ARC commands are Clear screen and Quit menu. The former option clears the terminal screen, and the latter option allows the user to end the session.

Table 2. Attribute list associated with selected ground-water-withdrawal sites, generated by the geographic information system

Regional number	Latitude	Longitude	Well- source identifier	Well depth (feet below land surface)
1	395725	743332	5200001	440
2	395720	743332	5200002	330
3	395635	743000	3215238	192
4	400437	742930	POND	10
5	400410	742858	POND	6
6	400430	742938	2809774	103
7	400240	742938	POND	6
8	400328	742855	POND	8
9	400325	742745	POND	10
10	400216	743607	2800796	1,055
11	400300	743514	2803943	802
12	400150	743428	2800795	1,094
13	400141	743525	2800797	1,024
14	395829	743506	3200526	298
15	395829	743503	3206929	315
16	395752	743452	3200386	334
17	395707	743246	3200383	345
18	395755	743239	3200818	338
19	395755	743239	3200819	332
20	400401	743200	4800022	240
21	400401	743201	2810933	241
22	395635	743000	MARKS RESV	2
23	395635	743000	UPPER CANAL PD	3
24	395635	743000	MIDDLE CANAL	3
25	400030	743600	4800268	980
26	400328	742742	POND	8
27	395845	743450	3212918	242
28	395730	743315	3212358	233
29	395723	743510	3212359	255
30	395718	743514	3212360	240
31	395635	743000	LOWER CANAL PD	3
32	395635	743000	BASIN RESV	15
33	395635	743000	UNION POND	1
34	395635	743000	ROME POND	15
35	395914	743329	3215968	364
36	400220	743518	PROPOSED	15
37	400200	742400	2902202	56
38	400130	742530	2803757	90
39	400121	742348	2908024	75
40	400130	742442	4800341	77
41	400126	742442	2911666	80

REFERENCES CITED

- Environmental Systems Research Institute, Inc., 1987a, Users guide ARC/INFO, 2 volumes: Redlands, Calif., Environmental Research Institute, Inc., unpaginated.
- ____ 1987b, ARC Macro Language and user interface tools: Redlands, Calif., Environmental Systems Research Institute, Inc., unpaginated.
- New Jersey Department of Environmental Protection, 1987, 1987 New Jersey Water Report: Trenton, N.J., New Jersey Department of Environmental Protection, 28 p.

Appendix 1. Source code for start.aml

```

/*-----
/* Program name: start.aml
/* Language: AML
/* Arc Version and Platform: 5.0, Prime
/* Subsystem: Arc
/*
/*:
/*
/* Purpose: This program begins the ARCPLLOT session and sets global
/*          variables. User pathnames can be changed in this program.
/*
/*:
/*
/* Arguments:
/* Variable name, Definition
/* -----
/* .ms, message environment
/* .tp, township boundary coverage
/* .cy, county boundary coverage
/* .pd, pinelands boundary coverage
/* .st, nj state boundary coverage
/* .str, 1:250,000 scale streams coverage
/* .sfc, surface-water-withdrawal points coverage
/* .gnd, ground-water-withdrawal points coverage
/* .path, pathname containing all files
/* .basin, drainage basins coverage
/* attfile, pathname of infofile
/* .conv, conversion factor
/*:
/*
/* Programs or menus called: design.menu
/*
/*:

&if [NULL [SHOW &station]] &then &do
    &if [NULL [SHOW &station]] &then &return /* quit
    &if [NULL [SHOW display]] &then &return Display not defined.
&end

&s .ms [SHOW &mess]
&ty Please wait...
&messages &off
ARCPLLOT
&messages &on;&dal 4

```

Appendix 1. Source code for start.aml--Continued

```
/* SET COVER PATHNAMES
&s .tp gis>lib>anth>civ>twp-250k
&s .cy gis>lib>anth>civ>cou-250k
&s .pd gis>lib>anth>luc>oth>pin
&s .st gis>lib>anth>civ>nj
&s .str gis>lib>anth>hyd>k250>str-250k
&s .sfc climate>pdunne>gis>menu>sw.cov
&s .gnd climate>pdunne>gis>menu>gw.cov
&s .path climate>pdunne>gis>menu
&s .basin climate>pdunne>gis>menu>basinam

/* set up relate between gw.cov and sw.cov point attribute tables to
/* WSOURCE file
&s attfile [dir %.gnd%]>INFO!ARC!WSOURCE
RELATE ADD;R;%attfile%;INFO;RECNUM;RECNUM;LINK;;

/*SET CONVERSION FACTOR TO CONVERT COVERAGE UNITS TO MILES
&s .conv 1610 /* feet per mile

/* DEFINE SYMBOL SETS FOR DISPLAY
LINESET color
MARKERSET color
/* ALLOW DRAWING TO EDGE OF MAP LIMITS
CLIPMAPEXTENT off

&menu DESIGN
&return
```

Appendix 2. Source code for design.menu

```
2 DESIGN DRAWING PARAMETERS...
/*-----
/* Program name: design.menu
/* Language: AML
/* Arc Version and Platform: 5.0, Prime
/* Subsystem: Arcplot
/*
/*:.....
/*
/* Purpose: This program displays the menu on the terminal screen, and invokes
/*          other programs after main menu options have been chosen.
/*
/*:.....
/* Other global variables: .xloc, .yloc, .radius, .gnd, .sfc, .ord
/*:.....
/*
/* Programs or menus called: llfind.aml, select.menu, order.aml, plot.aml,
/*                          zoomout.aml, plotfile.aml, calcgw.aml, glist.aml, calcs.aml,
/*                          slist.aml
/*
/* Programs or menus called by: start.aml
/*:.....

'Set site LAT,LON,RAD' &r llfind.aml
'Choose draw features' &menu select.menu &form
', '

'Draw (mapextent)'
'First order basin' &r order first
'Second order basin' &r order second
'Third order basin' &r order third
'Radius area' MAPEX [calc %.xloc% - %.radius%] [calc %.yloc% - ~
%.radius%] [calc %.xloc% + %.radius%] [calc %.yloc% + %.radius%];~
CLEAR;&r plot
'Zoom in/out'
'Zoom in box' MAPEX *; CLEAR;&r plot
'Zoom out' &r zoomout
'Set Mapscale'
automatic MAPSCALE auto
'1:24,000' MAPSCALE 24000
'Make a plotfile' &r plotfile
'Identify'
'GW site' IDENTIFY %.gnd% points * r//sourceid; &dal 4
'SW site' IDENTIFY %.sfc% points * r//sourceid r//locid
'Basin' IDENTIFY basinnam polys * %.ord%
'Refresh screen' CLEAR; &dal 4; &r plot
', '
```

Appendix 2. Source code for design.menu--Continued

```
'Select GW area'
'upstream polygon' CLEARSELECT; RESELECT %.gnd% point poly *; &r calcgw
'buffer area' CLEARSELECT;~
    RESELECT %.gnd% point circle %.xloc% %.yloc% %.radius%; &r calcgw
'mapextent' CLEARSELECT; RESELECT %.gnd% point mapex; &r calcgw
'clr all selections' CLEARSELECT
'Create GW data list'
'list1' &pause Data will be saved to file list.gw ...<cr> to continue...;~
    &r glist a
'list2' &pause Data will be saved to file list.gw ...<cr> to continue...;~
    &r glist b
'List GW data to screen'
'list1' LIST %.gnd% points R//LAT,R//LON,R//SOURCEID,R//LOCID,R//BASIN,~
    R//DEPTH,R//GEOL,R//CAPACITY, dist
'list2' LIST %.gnd% points R//LAT,R//LON,R//SOURCEID
'Select SW area'
'upstream polygon' CLEARSELECT; RESELECT %.sfc% point poly *; &r calcs
'buffer area' CLEARSELECT;~
    RESELECT %.sfc% point circle %.xloc% %.yloc% %.radius%; &r calcs
'mapextent' CLEARSELECT; RESELECT %.sfc% point mapex; &r calcs
'clr all selections' CLEARSELECT
'Create SW data list'
'list1' &pause Data will be saved to file list.sw ...<cr> to continue...;~
    &r slist a
'list2' &pause Data will be saved to file list.sw ...<cr> to continue...;~
    &r slist b
'List SW data to screen'
'list1' LIST %.sfc% points R//LAT,R//LON,R//SOURCEID,R//LOCID,R//BASIN,~
    R//DEPTH,R//GEOL,R//CAPACITY, dist
'list2' LIST %.sfc% points R//LAT,R//LON,R//SOURCEID
', '
'Clear Screen' CLEAR
'Quit menu' &ty '&menu design &force' to return to menu;&dal 0;&return
```

Appendix 3. Source code for 1lfind.aml

```

/*-----
/* Program name: 1lfind.aml
/* Language: AML
/* Arc Version and Platform: 5.0, Prime
/* Subsystem: Arcplot
/*
/*:.....
/*
/* Purpose: This program locates information for a selected site (latitude,
/*          longitude, radius). A latitude of 40 degrees, 34 minutes, 22
/*          seconds is entered as 403422. Longitudes are entered similarly.
/*          A 6.2-mile radius is entered as 6.2.
/*
/*:.....
/*
/* Arguments:
/* Variable name, Definition
/* -----
/* lat, latitude from user terminal input
/* lon, longitude from user terminal input
/* .rad, radius from user terminal input
/* .lt, latitude with spaces for proper format
/* .ln, longitude with spaces for proper format
/* tmp1, temporary file name
/* tmp2, temporary file name
/* ch, file unit number for I/O
/* st, status variable for I/O
/* xy, "real" world location
/* .xloc, location in UTM coordinates
/* .yloc, location in UTM coordinates
/* .radius, converts radius from miles to proper unit
/* .buf, "location set" flag
/*:.....
/*
/* Programs or menus called by: design.menu
/*
/*:.....

&dal 4
&messages &on

&s lat := [RESPONSE 'Enter latitude (ddmmss) (<cr> to quit)']
&if [null %lat%] &then &return Exiting...
&s lon := [RESPONSE 'Enter longitude (ddmmss)']
&if [null %lon%] &then &return Bad lon/lat value.
&s .rad := [response 'Enter radius (miles)']

```

Appendix 3. Source code for llfind.aml--Continued

```
&if [null %.rad%] &then &return Bad radius value.

/* extract lat/lon value (insert spaces for proper format)
&s .lt := [SUBSTR %lat% 1 2] [SUBSTR %lat% 3 2] [SUBSTR %lat% 5 2]
&s .ln := [SUBSTR %lon% 1 2] [SUBSTR %lon% 3 2] [SUBSTR %lon% 5 2]

/* check input
&ty %.lt% %.ln% %.rad%
&message &on

/* open temporary file for PROJECT and write lon/lat values
&s tmp1 := t$1[DATE -ftag]
&s tmp2 := t$2[DATE -ftag]
&s ch := [OPEN %tmp1% st -w]
&s st := [WRITE %ch% [QUOTE [UNQUOTE %.ln%] [UNQUOTE %.lt%]]]
&s st := [CLOSE %ch%]
&ty Projecting point into UTM ....
&sys ARC PROJECT FILE %tmp1% %tmp2% %.path%>DMS_UTM.PRJ /* DMS_STP.PRJ
                                                    /* for state plane

/* read state plane output from temp file
&s ch := [OPEN %tmp2% st -r]
&s xy := [UNQUOTE [READ %ch% st]]

/* close and delete temp files
&s st := [CLOSE %ch%]
&if [DELETE %tmp1%] ne 0 &then &ty Warning: Could not delete file %tmp1%
&if [DELETE %tmp2%] ne 0 &then &ty Warning: Could not delete file %tmp2%

&s .xloc := [EXTRACT 1 %xy%]
&s .yloc := [EXTRACT 2 %xy%]
&s .radius := [calc %.rad% * conv] /* convert miles to coverage units.
&s .buf .true.

&mess &on
```

Appendix 4. Source code for select.menu

```

/*-----
/* Program name: select.menu
/* Language: AML
/* Arc Version and Platform: 5.0, Prime
/* Subsystem: Arcplot
/*
/*:.....:
/*
/* Purpose: This program displays a menu on the terminal screen from which
/*          the user selects the features to be drawn.
/*:.....:
/*
/* Arguments:
/* Variable name, Definition
/* -----
/****** all variables are logical (yes/no) variables *****
/* .twmshp, draw township boundaries
/* .cnty, draw county boundaries
/* .pnlnd, draw pinelands boundaries
/* .strm, draw streams boundaries
/* .bsn, draw basin boundaries
/* .sfcw, draw surface-water-withdrawal points
/* .gndw, draw ground-water-withdrawal points
/*:.....:
/* Programs or menus called by: design.menu
/*:.....:

```

7 coverage menu

Select coverages to be drawn:
(Answer yes or no for ALL coverages)

COVERAGE	ON?
-----	---
Townships	%ON1
Counties	%ON2
Pinelands	%ON3
Streams (250k)	%ON4
Streams (100k)	%ON5
Basins	%ON6
Sfc. withdrawal pts.	%ON7
Gr.water w/draw pts.	%ON8

%OK

Appendix 4. Source code for select.menu--Continued

```
%ON1 input .twshp 8 next %ON2 choice yes no y n
%ON2 input .cnty 8 next %ON3 choice yes no y n
%ON3 input .pnlnd 8 next %ON4 choice yes no y n
%ON4 input .strm 8 next %ON5 choice yes no y n
%ON5 input .strms 8 next %ON6 choice yes no y n
%ON6 input .bsn 8 next %ON7 choice yes no y n
%ON7 input .sfcw 8 next %ON8 choice yes no y n
%ON8 input .gndw 8 next %OK choice yes no y n
%OK button return OK &return
%forminit
```

Appendix 5. Source code for order.aml

```

/*-----
/* Program name: order.aml
/* Language: AML
/* Arc Version and Platform: 5.0, Prime
/* Subsystem: Arcplot
/*
/*:
/*
/* Purpose: This program sets the mapextent and runs plot.aml. Basins will be
/*          drawn from the coverage identified in basinnam. The user can
/*          specify first, second, or third as an argument (.ord) in order to
/*          choose from three lists of basins (first order, second order, or
/*          third order).
/*
/*:
/*
/* Arguments:
/* Variable name, Definition
/* -----
/* bas, drainage basin to be displayed
/* .ord, basin order to be displayed, user supplied from menu, design.menu
/* basinnam, basin polygon coverage
/*:
/*
/* Programs or menus called: plot.aml
/*
/* Programs or menus called by: design.menu
/*:

&args .ord
&s bas [GETUNIQUE basinnam -polygon %ord% -none]
&if ^ [null %bas%] &then &do
    RESELECT basinnam poly %ord% = %bas%
    MAPEX poly basinnam
    CLEARSELECT
    CLEAR
    &r plot
&end

```

Appendix 6. Source code for plot.aml

```

/*-----
/* Program name: plot.aml
/* Language: AML
/* Arc Version and Platform: 5.0, Prime
/* Subsystem: Arcplot
/*
/*:.....
/*
/* Purpose: This program draws the drainage basins within the mapextent
/*          to the terminal screen.
/*
/*:.....
/* Global variables used: .twnshp, .cnty, .strm, .pnland, .sfcw, .gndw,
/*                        .bsn, .basin, .strms, .streams, .buf, .tp, .cy,
/*                        .str, .pd, .sfc, .gnd, .xloc, .yloc, .radius
/*:.....
/*
/* Programs or menus called by: design.menu, zoomout.aml, order.aml
/*
/*:.....

PAGESIZE 17 10
MAPLIMITS 4 0 17 10
MAPUNITS meters /* use feet for state plane coordinates, meters for utm
                /* coordinates
MAPPOSITION cen cen
LINESET color

/* plot to screen (using lineset color)
&if [SUBSTR %.twnshp% 1 1] = 'Y' &then ARCLINES %.tp% 1
&if [SUBSTR %.cnty% 1 1] = 'Y' &then ARCLINES %.cy% 1
&if [SUBSTR %.strm% 1 1] = 'Y' &then ARCLINES %.str% 5
&if [SUBSTR %.pnlnd% 1 1] = 'Y' &then ARCLINES %.pd% 3
&if [SUBSTR %.sfcw% 1 1] = 'Y' &then POINTMARKERS %.sfc% 18
&if [SUBSTR %.gndw% 1 1] = 'Y' &then POINTMARKERS %.gnd% 3
&if [SUBSTR %.bsn% 1 1] = 'Y' &then &do
    LINESYMBOL 8
    POLYGONS %.basin%
&end
&if [SUBSTR %.strms% 1 1] = 'Y' &then ARCLINES %.streams% 5

UNITS map
MARKERSET plotter.mrk
MARKERSYMBOL 69
&if %.buf% &then MARKER %.xloc% %.yloc%
LINESYMBOL 33
&if %.buf% &then CIRCLE %.xloc% %.yloc% %.radius%
MARKERSET color
UNITS page

```

Appendix 7. Source code for zoomout.aml

```

/*-----
/* Program name: zoomout.aml
/* Language: AML
/* Arc Version and Platform: 5.0, Prime
/* Subsystem: Arcplot
/*
/*:.....
/*
/* Purpose: This AML allows the user to resize a screen plot. To resize the
/*          screen plot, change the number in the &s .xdif and &s .ydif lines.
/*
/*:.....
/*
/* Arguments:
/* Variable name, Type, Definition
/* -----
/* .m, mapextent
/* .xmin, minimum "x" value of the mapextent
/* .ymin, minimum "y" value of the mapextent
/* .xmax, maximum "x" value of the mapextent
/* .ymax, maximum "y" value of the mapextent
/* .xdif, difference between "x" values times .2
/* .ydif, difference between "y" values times .2
/*:.....
/*
/* Programs or menus called: plot.aml
/*
/* Programs or menus called by: design.menu
/*:.....

&s .m [SHOW mapex]
&s .xmin [EXTRACT 1 %.m%]
&s .ymin [EXTRACT 2 %.m%]
&s .xmax [EXTRACT 3 %.m%]
&s .ymax [EXTRACT 4 %.m%]
&s .xdif = [CALC %.xmax% - %.xmin%] * .2
&s .ydif = [CALC %.ymax% - %.ymin%] * .2
MAPSCALE automatic
MAPEX [CALC %.xmin% - %.xdif%] [CALC %.ymin% - %.ydif%] ~
      [CALC %.xmax% + %.xdif%] [CALC %.ymax% + %.ydif%]
/* Refresh screen
CLEAR
&r plot

```

Appendix 8. Source code for plotfile.aml

```

/*-----
/* Program name: plotfile.aml
/* Language: AML
/* Arc Version and Platform: 5.0, Prime
/* Subsystem: Arcplot
/*
/*:.....:
/*
/* Purpose: This program determines the positions and characteristics of
/*          objects that are to be plotted on hardcopy
/*
/*:.....:
/*
/* Arguments:
/* Variable name, Type, Definition
/* -----
/* pf, plotfile pathname
/*:.....:
/*
/* Programs or menus called: plot2.aml
/*
/* Programs or menus called by: design.menu
/*:.....:

&s pf [RESPONSE 'enter plotfile name' #plt]
&mess &off
DISPLAY 1039
%pf%
&mess &on
&ty Creating plotfile...please wait...
PAGESIZE 15 15
MAPLIMITS 2.5 0.0 15.0 13.0
MAPUNITS meters /* use feet for state plane coor., meters for UTM.

/* draw legend
TEXTFONT 3
TEXTQUALITY proportional
TEXTSPACING 1.4 1.8
LINECOLOR 1
LINESIZE 0.02
LINETYPE wide
MAPPOS cen cen
BOX 0 0 15 15
BOX 2.5 0 15 13
MOVE 0.2 0.2

```

Appendix 8. Source code for plotfile.aml--Continued

```
TEXT 'Plotting software created by:'
MOVE 0.2 0.1
TEXT 'U.S.Geological Survey, W.Trenton, N.J. (9/91)'
MOVE 0.2 0.4
TEXT [QUOTE Created [DATE -vfull] by [USERNAME]]
LINECOLOR 3
LINE 0.5 0.75 1.0 0.75
MOVE 1.25 0.75
TEXT 'Pinelands boundary'
LINECOLOR 1
LINE 0.5 1.0 1.0 1.0
MOVE 1.25 1.0
TEXT Counties
LINE 0.5 1.25 1.0 1.25
MOVE 1.25 1.25
TEXT Townships
LINECOLOR 5
LINE 0.5 1.50 1.0 1.5
MOVE 1.25 1.5
TEXT Streams
LINECOLOR 8
LINE 0.5 1.75 1.0 1.75
MOVE 1.25 1.75
TEXT 'Basins'
MARKERSYMBOL 3
MARKER 0.75 2.0
MOVE 1.25 2.0
TEXT 'Ground water withdrawal'
MARKERSYMBOL 18
MARKER 0.75 2.25
MOVE 1.25 2.25
TEXT 'Surface withdrawal'
MARKERSET plotter.mrk
MARKERSYMBOL 69
MARKER 0.75 2.5
MOVE 1.25 2.5
TEXT 'Allocation site'
MOVE 0.75 2.75
TEXT [QUOTE &.rad& Mile radius indicated]
MAKESCALE oneway 0.2 7.0
MOVE 0.25 3.35
TEXT 'This legend lists all possible items that can'
MOVE 0.25 3.25
TEXT 'be plotted. This plot only contains those'
MOVE 0.25 3.15
TEXT 'items that have been selected.'
MOVE 0.5 6.4
TEXT [QUOTE Scale: 1:[ROUND [SHOW mapscale]]]
```

Appendix 8. Source code for plotfile.aml--Continued

```
MARKERSET municipal.mrk
MARKERSIZE 1.0
MARKERSYMBOL 128
MARKER 1.25 8.0
MOVE 7.5 14
TEXTSIZE 0.25
TEXT 'Relation of Proposed Allocation Site to Existing Water Withdrawal ~
Points' CC
TEXTSIZE 0
MOVE 0.5 12.5
TEXT [QUOTE Points are located within %.rad% miles of:]
MOVE 0.5 12
TEXT [QUOTE LATITUDE %.1t%]
MOVE 0.5 11.8
TEXT [QUOTE LONGITUDE %.1n%]
MARKERSET color

/* plots to printer
&r plot2
&dal 0
display 0
&mess &popup
&if [QUERY 'sendplot']~
    &then sys arc plot %pf%
&sta [SHOW &sta]
&mess &on
&dal 4
```

Appendix 9. Source code for plot2.aml

```

/*-----
/* Program name: plot2.aml
/* Language: AML
/* Arc Version and Platform: 5.0, Prime
/* Subsystem: Arcplot
/*
/*:.....
/*
/* Purpose: This program creates a hardcopy plot and specifies colors for
/*          the plotter
/*
/*:.....
/* Other global variables: .twshp, .cnty, .strm, .pnlnd, .sfcw, .gndw,
/*                          .bsn, .basin, .strms, .streams, .buf, .tp, .cy,
/*                          .str, .pd, .sfc, .gnd, .xloc, .yloc, .radius
/* -----
/*
/*:.....
/*
/* Programs or menus called by: plotfile.aml
/*
/*:.....

LINESET color
/* plot MAP (colors for plotter) (using lineset color)
&if [SUBSTR %.twshp% 1 1] = 'Y' &then ARCLINES %.tp% 1
&if [SUBSTR %.cnty% 1 1] = 'Y' &then ARCLINES %.cy% 1
&if [SUBSTR %.strm% 1 1] = 'Y' &then ARCLINES %.str% 5
&if [SUBSTR %.pnlnd% 1 1] = 'Y' &then ARCLINES %.pd% 3
&if [SUBSTR %.sfcw% 1 1] = 'Y' &then POINTMARKERS %.sfc% 18
&if [SUBSTR %.gndw% 1 1] = 'Y' &then POINTMARKERS %.gnd% 3
&if [SUBSTR %.bsn% 1 1] = 'Y' &then &do
    LINESYMBOL 8
    POLYGONS %.basin%
&end
&if [SUBSTR %.strms% 1 1] = 'Y' &then ARCLINES %.streams% 5

UNITS map
MARKERSET plotter.mrk
MARKERSYMBOL 69
&if %.buf% &then MARKER %.xloc% %.yloc%
LINESYMBOL 33
&if %.buf% &then CIRCLE %.xloc% %.yloc% %.radius%
MARKERSET color
UNITS page

```


Appendix 10. Source code for calcgw.aml and calcsw.aml

```

/*-----
/* Program names: (a)calcgw.aml, (b)calcsw.aml
/* Language: AML
/* Arc Version and Platform: 5.0, Prime
/* Subsystem: Arcplot
/*
/*:.....
/*
/* Purpose: These programs calculate distances between ground-water- or
/*          surface-water-withdrawal sites and the proposed allocation site,
/*          within a specified area.
/*
/*:.....
/*
/* Arguments:
/* Variable name, Type, Definition
/* -----
/* dist, distance of ground water sites from proposed allocation point
/* x-coord, UTM location of selected set
/* y-coord, UTM location of selected set
/* -----
/* Other global variables: .conv, .xloc, .yloc
/*:.....
/*
/* Programs or menus called by: design.menu
*
/*:.....

/* program (a)---calcgw.aml

CALCULATE GW.COV POINTS DIST = ( ( %.XLOC% - X-COORD ) * ( %.XLOC% - X-COORD )~
) + ( ( %.YLOC% - Y-COORD ) * ( %.YLOC% - Y-COORD ) )
CALCULATE GW.COV POINTS DIST = DIST ** 0.5
CALCULATE GW.COV POINTS DIST = DIST / %.CONV%

/* program (b)---calcsw.aml
/*
CALCULATE SW.COV POINTS DIST = ( ( %.XLOC% - X-COORD ) * ( %.XLOC% - X-COORD )~
) + ( ( %.YLOC% - Y-COORD ) * ( %.YLOC% - Y-COORD ) )
CALCULATE SW.COV POINTS DIST = DIST ** 0.5
CALCULATE SW.COV POINTS DIST = DIST / %.CONV%

```

Appendix 11. Source code for glist.aml and slist.aml

```

/*-----
/* Program names: (a)glist.aml, (b)slist.aml
/* Language: AML
/* Arc Version and Platform: 5.0, Prime
/* Subsystem: Arcplot
/*
/*:-----
/*
/* Purpose: These programs allow the user to produce a list of attributes
/*          for an area surrounding ground-water- or surface-water-
/*          withdrawal sites. The user can place attribute lists that are
/*          to be printed after the INFOFILE command. If it is necessary to
/*          expand the print options, add another &when command and the
/*          corresponding INFOFILE command containing the attribute list.
/*
/*:-----
/*
/* Arguments:
/* Variable name, Type, Definition
/* -----
/* pr, choice flag, supplied by design.menu
/* -----
/* Other global variables: .gnd, .sfc
/*:-----
/*
/* Programs or menus called by: design.menu
/*
/*:-----

/* program (a)---glist.aml
&args pr
&select %pr%
  &when a
    INFOFILE %.GND% POINTS TMP R//LAT,R//LON,~
    R//SOURCEID,R//DEPTH
  &when b
    INFOFILE %.gnd% POINTS TMP R//LAT,R//LON,R//SOURCEID
  &end
&data ARC INFO
OUTPUT [path LIST.GW] INIT
SELECT TMP
LIST PRINT
ERASE TMP
Y
Q STOP
&end

```

Appendix 11. Source code for glist.aml and slist.aml--Continued

```
/*program (b)---slist.aml
&args pr
&select %pr%
  &when a
    INFOFILE %.sfc% POINTS TMP R//LAT,R//LON,R//SOURCEID,R//LOCID,~
      R//BASIN,R//DEPTH,R//GEO1,R//CAPACITY INIT
  &when b
    INFOFILE %.sfc% POINTS TMP R//LAT,R//LON,R//SOURCEID
  &end
&data ARC INFO
OUTPUT [path LIST.SW] INIT
SELECT TMP
LIST PRINT
ERASE TMP
Y
Q STOP
&end
```